BB8: SVD

Eric Miller - QEA - April 13, 2016

Definitions

```
In[1]:= SetDirectory[NotebookDirectory[]];
```

Eventually, these functions may want to use the Notation` package to actually make the subscripted letters have downvalues. For now, I'm just being careful.

```
In[2]:= Clear@ClearSubscripts
ClearSubscripts[top_] := (
   (Unset[Evaluate[#[[1]]]];
      Evaluate[#[[1]]]) & /@ Select[DownValues[Subscript], #[[1, 1, 1]] == top &]
  )
Clear@unknownArray
unknownArray[top_, dim_ /; NumberQ[dim]] := (
   Table[top_i, {i, dim}]
  )
unknownArray[top_, {dim1_ /; NumberQ[dim1], dim2_ /; NumberQ[dim2]}] := (
   Table[top_{i,j}, {i, dim1}, {j, dim2}]
  )
```

Eigenvectors of Data

Because we are defining the eigenvectors (including v_M) to be normalized, $|| v_M || = || w || = 1$.

$$u = \frac{w}{|| w ||} = w$$

Equation 7 in the packet states that

$$\sigma_w{}^2 = \frac{w^T R w}{w^T w} \tag{1}$$

The fact that w is an eigenvector means $R w = \lambda_M w$.

$$\sigma_w{}^2 = \frac{w^T \lambda_M w}{w^T w} \tag{2}$$

The commutivity of scalar multiplication implies

$$\sigma_w{}^2 = \lambda_M \frac{w^T w}{w^T w} = \lambda_M \tag{3}$$

QED

SVD Image Compression

Compression

Without looking at the provided sample code at all, I decided to write my own in *Mathematica*. It turns out we were *supposed* to do that. Oops.

```
mark = ColorConvert[
```

```
, "Grayscale"]
```



```
In[7]:= compress[n_, image_] := Module[{u, w, v, u2, w2, v2},
  {u, w, v} = SingularValueDecomposition[ImageData@image];
  u2 = u[[All, 1 ;; n]];
  w2 = w[[1 ;; n, 1 ;; n]];
  v2 = v[[All, 1 ;; n]];
  Image[u2.w2.Transpose@v2]
 ]
```

```
Grid[Table[{i, Image[compress[i, mark], ImageSize → Tiny]},
  {i, {1, 3, 5, 10, 20, 50, 100, 200}}]]
```

1	
3	
5	
10	
20	
50	
100	
200	

Exercises

$$A = \begin{pmatrix} 3 & 2 \\ 2 & 3 \\ 2 & -2 \end{pmatrix};$$

```
Grid@Eigensystem[A.Transpose@A]
Grid@Eigensystem[Transpose[A].A]
```

25	9	0
{1, 1, 0}	{1, -1, 4}	{-2, 2, 1}

25	9
{1, 1}	{-1, 1}

```
MatrixForm[vt = Normalize /@ Eigenvectors[Transpose@A.A]]
vt[[All, 1]].vt[[All, 2]]
```

$$\text{sigma = Sqrt@}\begin{pmatrix} 25 & 0 \\ 0 & 9 \end{pmatrix};$$

```
MatrixForm[u = Transpose[Normalize /@ Eigenvectors[A.Transpose@A][[1 ;; 2]] * {1, -1}]]
u[[All, 1]].u[[All, 2]]
```

$$\begin{pmatrix} \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \\ -\frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \end{pmatrix}$$

0

$$\begin{pmatrix} \frac{1}{\sqrt{2}} & -\frac{1}{3\sqrt{2}} \\ \frac{1}{\sqrt{2}} & \frac{1}{3\sqrt{2}} \\ 0 & -\frac{2\sqrt{2}}{3} \end{pmatrix}$$

0

```
MatrixForm /@ {A, u.sigma.vt}
```

$$\left\{\begin{pmatrix} 3 & 2 \\ 2 & 3 \\ 2 & -2 \end{pmatrix}, \begin{pmatrix} 3 & 2 \\ 2 & 3 \\ 2 & -2 \end{pmatrix}\right\}$$

SVD

1.a.

$$A = \begin{pmatrix} 1 & 1 \\ 0 & 1 \\ -1 & 1 \end{pmatrix};$$

```
Grid@Eigensystem[A.Transpose@A]
Grid@Eigensystem[Transpose@A.A]
```

3	2	0
{1, 1, 1}	{-1, 0, 1}	{1, -2, 1}

3	2
{0, 1}	{1, 0}

Now, find the major components.

```
eig1 = Normalize /@ Eigenvectors[A.Transpose@A, 2] * {1, -1};
u = Transpose@eig1;
HoldForm@u == (u // MatrixForm)

lambda = DiagonalMatrix@Sqrt@Eigenvalues[Transpose@A.A];
HoldForm@Λ == (lambda // MatrixForm)

eig2 = Normalize /@ Eigenvectors[Transpose@A.A, 2];
v = Transpose@eig2;
HoldForm@v == (v // MatrixForm)
```

$$u == \begin{pmatrix} \frac{1}{\sqrt{3}} & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{3}} & 0 \\ \frac{1}{\sqrt{3}} & -\frac{1}{\sqrt{2}} \end{pmatrix}$$

$$\Lambda == \begin{pmatrix} \sqrt{3} & 0 \\ 0 & \sqrt{2} \end{pmatrix}$$

$$v == \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}$$

Finally, verify that the solution works.

```
HoldForm@A == MatrixForm@A == MatrixForm[u.lambda.Transpose[v]]
A == u.lambda.Transpose[v]
```

$$A == \begin{pmatrix} 1 & 1 \\ 0 & 1 \\ -1 & 1 \end{pmatrix} == \begin{pmatrix} 1 & 1 \\ 0 & 1 \\ -1 & 1 \end{pmatrix}$$

True

1.b.

$$A = \begin{pmatrix} 3 & 2 & 2 \\ 2 & 3 & -2 \end{pmatrix};$$

```
Grid@Eigensystem[A.Transpose@A]
Grid@Eigensystem[Transpose@A.A]
```

25	9
{1, 1}	{-1, 1}

25	9	0
{1, 1, 0}	{1, -1, 4}	{-2, 2, 1}

Now, find the major components.

```
eig1 = Normalize /@ Eigenvectors[A.Transpose@A, 2] * {1, -1};
u = Transpose@eig1;
HoldForm@u == (u // MatrixForm)

lambda = DiagonalMatrix@Sqrt@Eigenvalues[Transpose@A.A, 2];
HoldForm@Λ == (lambda // MatrixForm)

eig2 = Normalize /@ Eigenvectors[Transpose@A.A, 2];
v = Transpose@eig2;
HoldForm@v == (v // MatrixForm)
```

$$u == \begin{pmatrix} \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & -\frac{1}{\sqrt{2}} \end{pmatrix}$$

$$\Lambda == \begin{pmatrix} 5 & 0 \\ 0 & 3 \end{pmatrix}$$

$$v == \begin{pmatrix} \frac{1}{\sqrt{2}} & \frac{1}{3\sqrt{2}} \\ \frac{1}{\sqrt{2}} & -\frac{1}{3\sqrt{2}} \\ 0 & \frac{2\sqrt{2}}{3} \end{pmatrix}$$

Finally, verify that the solution works.

```
HoldForm@A == MatrixForm@A == MatrixForm[u.lambda.Transpose[v]]
A == u.lambda.Transpose[v]
```

$$A == \begin{pmatrix} 3 & 2 & 2 \\ 2 & 3 & -2 \end{pmatrix} == \begin{pmatrix} 3 & 2 & 2 \\ 2 & 3 & -2 \end{pmatrix}$$

```
True
```

2.

When I apply SingularValueDecomposition[], the result tends to have an extra zero row floating around. I think it's from the *reduced* thing hinted at earlier.

$$\texttt{MatrixForm /@ SingularValueDecomposition@}\begin{pmatrix} 1 & 1 \\ 0 & 1 \\ -1 & 1 \end{pmatrix}$$

$$\texttt{MatrixForm /@ SingularValueDecomposition@}\begin{pmatrix} 3 & 2 & 2 \\ 2 & 3 & -2 \end{pmatrix}$$

$$\left\{ \begin{pmatrix} \frac{1}{\sqrt{3}} & \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{6}} \\ \frac{1}{\sqrt{3}} & 0 & -\sqrt{\frac{2}{3}} \\ \frac{1}{\sqrt{3}} & -\frac{1}{\sqrt{2}} & \frac{1}{\sqrt{6}} \end{pmatrix}, \begin{pmatrix} \sqrt{3} & 0 \\ 0 & \sqrt{2} \\ 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix} \right\}$$

$$\left\{ \begin{pmatrix} \frac{1}{\sqrt{2}} & -\frac{1}{\sqrt{2}} \\ \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \end{pmatrix}, \begin{pmatrix} 5 & 0 & 0 \\ 0 & 3 & 0 \end{pmatrix}, \begin{pmatrix} \frac{1}{\sqrt{2}} & -\frac{1}{3\sqrt{2}} & -\frac{2}{3} \\ \frac{1}{\sqrt{2}} & \frac{1}{3\sqrt{2}} & \frac{2}{3} \\ 0 & -\frac{2\sqrt{2}}{3} & \frac{1}{3} \end{pmatrix} \right\}$$

Applications

Fibbonacci

Setup the givens in the problem

```
A = (1 1
     1 0);
U0 = (1
      1);
```

$$A = \begin{pmatrix} 1 & 1 \\ 1 & 0 \end{pmatrix};$$
$$U0 = \begin{pmatrix} 1 \\ 1 \end{pmatrix};$$

Find the Eigendecomposition

```
{vals, vecs} = Eigensystem@A;
(Q = Transpose@vecs) // MatrixForm
(Λ = DiagonalMatrix@vals) // MatrixForm
```

$$\begin{pmatrix} \frac{1}{2}\left(1+\sqrt{5}\right) & \frac{1}{2}\left(1-\sqrt{5}\right) \\ 1 & 1 \end{pmatrix}$$

$$\begin{pmatrix} \frac{1}{2}\left(1+\sqrt{5}\right) & 0 \\ 0 & \frac{1}{2}\left(1-\sqrt{5}\right) \end{pmatrix}$$

```
Simplify[Q.MatrixPower[Λ, 39].Inverse@Q.U0] // MatrixForm
```

$$\begin{pmatrix} 165\,580\,141 \\ 102\,334\,155 \end{pmatrix}$$

```
Fibonacci[40]
```

102 334 155

```
Simplify[Q.MatrixPower[Λ, 99].Inverse@Q.U0] // MatrixForm
```

$$\begin{pmatrix} 573\,147\,844\,013\,817\,084\,101 \\ 354\,224\,848\,179\,261\,915\,075 \end{pmatrix}$$

```
Fibonacci[100]
```

354 224 848 179 261 915 075

Temperature Compression

```
In[8]:= Clear[bTr, wTr, sTr, bNew, sNew, wNew]
```

```
In[9]:= {wTr, wNew, bTr, bNew, sNew, sTr, pNew, pTr} =
          Import["avg_temperatures_pt2.mat"][[All, All, 1]];
```

Analyze the input data

```
In[11]:= ListPointPlot3D[Transpose@{bTr, wTr, sTr},
          AxesLabel → {"Boston", "Washington", "Sao Paolo"}]
```

Out[11]=



Build a covariance matrix

```
In[12]:= R = Covariance[Transpose@{bTr, wTr, sTr}];
         R // MatrixForm
         Grid@N@Round[Eigensystem@R, 10^-2]
```

Out[13]//MatrixForm=

$$\begin{pmatrix} 281.993 & 267.413 & -57.3231 \\ 267.413 & 282.566 & -57.5062 \\ -57.3231 & -57.5062 & 39.7308 \end{pmatrix}$$

Out[14]=

562.31	27.12	14.87
{-0.7, -0.7, 0.15}	{0.11, 0.1, 0.99}	{0.71, -0.71, -0.01}

```
In[15]:= Vp = Transpose@Eigenvectors[R, 2];
         Vp // MatrixForm
```

Out[16]//MatrixForm=

$$\begin{pmatrix} -0.698331 & 0.11341 \\ -0.699114 & 0.103721 \\ 0.153535 & 0.988119 \end{pmatrix}$$

Start working toward compression

```
In[27]:= T = # - Mean@# & /@ {bNew, wNew, sNew};
         T[[All, 1 ;; 5]] // MatrixForm
```

Out[28]//MatrixForm=

$$\begin{pmatrix} -25.3578 & -17.0578 & -23.1578 & -13.1578 & -11.8578 \\ -24.7814 & -18.2814 & -20.6814 & -10.1814 & -13.8814 \\ 10.7401 & 11.0401 & 4.34014 & 3.74014 & 3.34014 \end{pmatrix}$$

Do the compression

```
In[29]:= compressed = Transpose[Vp].T;
         compressed[[All, 1 ;; 5]] // MatrixForm
```

Out[30]//MatrixForm=

$$\begin{pmatrix} 36.6821 & 26.3878 & 31.2968 & 16.8807 & 18.4982 \\ 5.16635 & 7.07828 & -0.482852 & 1.14745 & 0.515867 \end{pmatrix}$$

Decompress

In[31]:=
```
reconstructed = Vp.compressed;
reconstructed[[All, 1 ;; 5]] // MatrixForm
```

Out[32]//MatrixForm=

$$\begin{pmatrix} -25.0304 & -17.6247 & -21.9103 & -11.6582 & -12.8593 \\ -25.1092 & -17.7139 & -21.9301 & -11.6825 & -12.8788 \\ 10.737 & 11.0456 & 4.32804 & 3.72559 & 3.34985 \end{pmatrix}$$

Analyze the results

In[33]:=
```
(* Many apologies for the cluttered appearance of this code. I promise it
 isn't that bad. *)
cities = {"Boston", "Washington", "Sao Paolo"};
Grid[
 Table[{ListLinePlot[{T[[i]], reconstructed[[i]]}, PlotStyle → {Blue, Green},
     PlotMarkers → None, PlotLabel → cities[[i]], ImageSize → 350],
   CorrelationDistance[T[[i]], reconstructed[[i]]]}, {i, 3}], Spacings → {0, 3},
 Frame → {False, All}]
```

Out[34]=



0.0135384



0.0148274



0.0000131709

Misc