Eigenthings pt 1

Eric Miller - QEA - April 06, 2016

Definitions

Eigenvalues and Eigenvectors of a Diagonal Matrix

2a.

```
(A = DiagonalMatrix[{-3, -1, 4}]) // MatrixForm
```

$$\begin{pmatrix} -3 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 4 \end{pmatrix}$$

```
Grid[{{{1, 0, 0} // MatrixForm, -3},
  {{0, 1, 0} // MatrixForm, -1}, {{0, 0, 1} // MatrixForm, 4}}, Frame → All]
```

$\begin{pmatrix} 1 \\ 0 \\ 0 \end{pmatrix}$	-3
$\begin{pmatrix} 0 \\ 1 \\ 0 \end{pmatrix}$	-1
$\begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix}$	4

2b.

```
(A = DiagonalMatrix[{1, 1, 2}]) // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 2 \end{pmatrix}$$

```
Grid[{{{1, 0, 0} // MatrixForm, 1},
  {{0, 1, 0} // MatrixForm, 1}, {{0, 0, 1} // MatrixForm, 2}}, Frame → All]
```

$\begin{pmatrix}1\\0\\0\end{pmatrix}$	1
$\begin{pmatrix}0\\1\\0\end{pmatrix}$	1
$\begin{pmatrix}0\\0\\1\end{pmatrix}$	2

2c.

```
(A = DiagonalMatrix[{2, 4, 0}]) // MatrixForm
```

$$\begin{pmatrix}2 & 0 & 0\\0 & 4 & 0\\0 & 0 & 0\end{pmatrix}$$

```
Grid[
 Table[{MatrixForm@UnitVector[Length@A, i], A[[i, i]]}, {i, Length@A}], Frame → All]
```

$\begin{pmatrix}1\\0\\0\end{pmatrix}$	2
$\begin{pmatrix}0\\1\\0\end{pmatrix}$	4
$\begin{pmatrix}0\\0\\1\end{pmatrix}$	0

Eigenvalues of a Triangular Matrix

Solve$\left[\begin{pmatrix}1 & 1\\0 & 1\end{pmatrix}.\{x, y\} == \{x, y\}, \{x, y\}\right]$

Solve::svars : Equations may not give solutions for all "solve" variables. >>

{{y → 0}}

Roots$\left[\text{Det}\left[\begin{pmatrix}a & b\\0 & c\end{pmatrix} - \lambda\,\text{IdentityMatrix[2]}\right] == 0, \lambda\right]$

$\lambda == c \,||\, \lambda == a$

3.

$$\begin{bmatrix} 2 & 1 \\ 0 & -3 \end{bmatrix} \begin{bmatrix} 1/5 \\ -1 \end{bmatrix} = \begin{bmatrix} 2/5 - 1 \\ 0 + 3 \end{bmatrix} = \begin{bmatrix} -3/5 \\ 3 \end{bmatrix} = -3 \cdot \begin{bmatrix} 1/5 \\ -1 \end{bmatrix}$$

4.

I decided to automate here, and gained a ton of insight from doing so. In particular, thinking about solving the equation $A.x = \lambda.x$ for known λ as looking for the null space of a matrix is really cool.

In[7]:=
```
triangleEigens[A_] := Module[{},
  values = Diagonal[A];
  vectors = Table[MatrixForm@
     Transpose@NullSpace[A - val IdentityMatrix[Length@A]], {val, values}];
  Grid[Transpose[{values,
     vectors}], Frame → All]
 ]
```

triangleEigens$\left[\begin{pmatrix} 1 & 0 \\ 1 & 2 \end{pmatrix}\right]$

1	$\begin{pmatrix} -1 \\ 1 \end{pmatrix}$
2	$\begin{pmatrix} 0 \\ 1 \end{pmatrix}$

triangleEigens$\left[\begin{pmatrix} 1 & 0 & 0 \\ -1 & 2 & 0 \\ 2 & 1 & 3 \end{pmatrix}\right]$

1	$\begin{pmatrix} -2 \\ -2 \\ 3 \end{pmatrix}$
2	$\begin{pmatrix} 0 \\ -1 \\ 1 \end{pmatrix}$
3	$\begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix}$

triangleEigens$\left[\begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 1 & 4 \end{pmatrix}\right]$

2	$\begin{pmatrix} 0 & 1 \\ -2 & 0 \\ 1 & 0 \end{pmatrix}$
2	$\begin{pmatrix} 0 & 1 \\ -2 & 0 \\ 1 & 0 \end{pmatrix}$
4	$\begin{pmatrix} 0 \\ 0 \\ 1 \end{pmatrix}$

This last problem is particularly interesting because the value 2 appears multiple times on the diagonal, and is thus a repeated eigenvalue. There *are* still three distinct eigenvectors, but pairing them up nicely is difficult.

The Characteristic Equation

Nice explanation! When I first learned this, it seemed unmotivated, but seeing the derivation this plainly makes it obvious.

```
In[8]:= myCharacteristicEqn[A_] := Module[{},
          Factor@Det[A - λ IdentityMatrix[Length@A]] == 0
        ]
```

Diagonal matrices

```
myCharacteristicEqn[DiagonalMatrix[{-3, -1, 4}]]
```

$-(-4+\lambda)(1+\lambda)(3+\lambda) == 0$

I don't know why exactly the leading - snuck in, but it isn't particuarly important.

```
myCharacteristicEqn[DiagonalMatrix[{1, 1, 2}]]
```

$-(-2+\lambda)(-1+\lambda)^2 == 0$

As expected, there is a squared term representing the duplicate eigenvalue.

```
myCharacteristicEqn[DiagonalMatrix[{2, 4, 0}]]
```

$-(-4+\lambda)(-2+\lambda)\lambda == 0$

And bare eigenvalues of 0 work too!

Triangular Matrices

$\texttt{myCharacteristicEqn}\left[\begin{pmatrix} 1 & 0 \\ 1 & 2 \end{pmatrix}\right]$

$(-2+\lambda)(-1+\lambda) == 0$

$\texttt{myCharacteristicEqn}\left[\begin{pmatrix} 1 & 0 & 0 \\ -1 & 2 & 0 \\ 2 & 1 & 3 \end{pmatrix}\right]$

$-(-3+\lambda)(-2+\lambda)(-1+\lambda) == 0$

$\texttt{myCharacteristicEqn}\left[\begin{pmatrix} 2 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 1 & 4 \end{pmatrix}\right]$

$-(-4+\lambda)(-2+\lambda)^2 == 0$

8.

Because A is triangular, $A - \lambda \mathrm{I}$ must also be diagonal. We showed earlier that this implies that det($A - \lambda$I) is the product of the diagonal elements of that matrix, which are first-degree polynomials of the form ($a - \lambda$) where $a \in$ Diagonal(A). Thus, their product is a degree-n polynomial with roots consisting of the diagonal elements of *A*.

7.

Diagonal matrices are triangular. See problem 8.

Eigenvalues and Eigenvectors of 2x2

9.

```
In[22]:= A = (18  -2
              12   7);
         15 → MatrixForm@NullSpace[A - 15 IdentityMatrix[2]][[1]]
         10 → MatrixForm@NullSpace[A - 10 IdentityMatrix[2]][[1]]
```

Out[23]= $15 \to \begin{pmatrix} 2 \\ 3 \end{pmatrix}$

Out[24]= $10 \to \begin{pmatrix} 1 \\ 4 \end{pmatrix}$

10.

```
In[34]:= A = (3  -2
              4  -1);
         HoldForm@Det[A] == A[[1, 1]] * A[[2, 2]] - A[[1, 2]] * A[[2, 1]]
         HoldForm@Tr[A] == Total@Diagonal@A
         TraditionalForm[λ^2 - Tr[A] λ + Det[A] == 0]
```

Out[35]= Det[A] == 5

Out[36]= Tr[A] == 2

Out[37]//TraditionalForm=

$$\lambda^2 - 2\,\lambda + 5 = 0$$

Relatively standard

```
In[38]:= A = (0  -1
              1   0);
         HoldForm@Det[A] == A[[1, 1]] * A[[2, 2]] - A[[1, 2]] * A[[2, 1]]
         HoldForm@Tr[A] == Total@Diagonal@A
         TraditionalForm[λ^2 - Tr[A] λ + Det[A] == 0]
```

Out[39]= Det[A] == 1

Out[40]= Tr[A] == 0

Out[41]//TraditionalForm=

$$\lambda^2 + 1 = 0$$

This equation has no real roots.

```
In[46]:= A = ( Cos[θ]  -Sin[θ]
               Sin@θ   Cos@θ  );
HoldForm@Det[A] == A[[1, 1]] * A[[2, 2]] - A[[1, 2]] * A[[2, 1]]
HoldForm@Tr[A] == Total@Diagonal@A
TraditionalForm[Simplify[λ^2 - Tr[A] λ + Det[A]] == 0]

Out[47]= Det[A] == Cos[θ]^2 + Sin[θ]^2

Out[48]= Tr[A] == 2 Cos[θ]

Out[49]//TraditionalForm=
```

$$-2\lambda\cos(\theta) + \lambda^2 + 1 = 0$$

After throwing on a simplify, lots of the trig goes away, which is nice.

11.

```
In[140]:= myEigenValues[A_] := (
   Clear@λ;
   List @@ Roots[myCharacteristicEqn[A], λ][[All, 2]]
  )

In[163]:= myEigenValues@( 3  -2
                          4  -1 )

Out[163]= {1 - 2 ⅈ, 1 + 2 ⅈ}

In[164]:= myEigenValues@( 0  -1
                          1   0 )

Out[164]= {ⅈ, -ⅈ}

In[170]:= myEigenValues@( Cos[θ]  -Sin[θ]
                          Sin@θ   Cos@θ  )
% == myEigenValues@RotationMatrix@θ

Out[170]= {Cos[θ] - ⅈ Sin[θ], Cos[θ] + ⅈ Sin[θ]}

Out[171]= True
```

12.

There is an awesome theorem showing that the sum of the roots of any polynomial in standard form is always $-\frac{b}{a}$ where b is the coefficient on the x^{n-1} term and a is the coefficient on the x^n term. In this case, $b = -\det(A) \wedge a = 1$, so the sum of the roots (the sum of the eigenvalues) must be $\det(A)$.

13.

A second related theorem shows that the product of the roots of any polynomial is $(-1)^n * \frac{z}{a}$ where n is the degree of the polynomial, and z is the constant term. In this case, $n = 2 \wedge z = \det(A)$, so the product of the roots is $\det(A)$.

14.

Consider $A = \begin{pmatrix} a & b \\ b & d \end{pmatrix}$. $\det(A) = a\,d - b^2$. $\mathrm{tr}(a) = a + d$. In order for the eigenvalues to be complex, it is necessarily true that $\mathrm{tr}(A)^2 - 4\det(A) < 0$. Expanding, $(a+d)^2 - 4\left(-b^2 + a\,d\right) < 0$, or $a^2 + 4\,b^2 - 2\,a\,d + d^2 < 0$. In the worst case, $b = 0$, so $a^2 - 2\,a\,d + d^2 = (a-d)^2 < 0$. Because $a - d$ is real, its square must be non-negative, so the discriminant of the equation must be nonnegative, and the eigenvalue(s) must be real.

In[57]:=
```
A = {{a, b}, {b, d}};
```

In[83]:=
```
Tr[A]^2 - 4 Det[A] < 0
Expand@%
% /. b → 0
Factor@%
```

Out[83]= $(a + d)^2 - 4\left(-b^2 + a\,d\right) < 0$

Out[84]= $a^2 + 4\,b^2 - 2\,a\,d + d^2 < 0$

Out[85]= $a^2 - 2\,a\,d + d^2 < 0$

Out[86]= $(a - d)^2 < 0$

Or, the lame way,

In[88]:=
```
Reduce[Tr[A]^2 - 4 Det[A] < 0]
```

Out[88]= False

15.

As discovered Nathan and Sam, the problem here should read $v_2 = \begin{pmatrix} 1 \\ i \end{pmatrix}$.

In[123]:=
```
A = (1  -1
     1   1);
λ2 = 1 - i;
v2 = (1
      i);
A.v2 // MatrixForm
λ2 v2 // MatrixForm
```

Out[126]//MatrixForm=
$$\begin{pmatrix} 1 - i \\ 1 + i \end{pmatrix}$$

Out[127]//MatrixForm=
$$\begin{pmatrix} 1 - i \\ 1 + i \end{pmatrix}$$

16.

This uses definitions from problem 11.

In[151]:=
```
myEigenVectors[A_] := (
  DeleteDuplicates@
   Flatten[Map[NullSpace[A - # IdentityMatrix[Length@A]] &, myEigenValues[A]], 1]
 )
```

In[155]:= **MatrixForm /@ myEigenVectors @ $\begin{pmatrix} 3 & -2 \\ 4 & -1 \end{pmatrix}$**

Out[155]= $\left\{\begin{pmatrix} 1 - i \\ 2 \end{pmatrix}, \begin{pmatrix} 1 + i \\ 2 \end{pmatrix}\right\}$

In[156]:= **MatrixForm /@ myEigenVectors @ $\begin{pmatrix} 0 & -1 \\ 1 & 0 \end{pmatrix}$**

Out[156]= $\left\{\begin{pmatrix} i \\ 1 \end{pmatrix}, \begin{pmatrix} -i \\ 1 \end{pmatrix}\right\}$

In[160]:= **MatrixForm /@ myEigenVectors @ $\begin{pmatrix} \text{Cos}[\theta] & -\text{Sin}[\theta] \\ \text{Sin}@\theta & \text{Cos}@\theta \end{pmatrix}$**

```
MatrixForm /@ myEigenVectors @ RotationMatrix[θ]
myEigenValues @ RotationMatrix[θ]
```

Out[160]= $\left\{\begin{pmatrix} -i \\ 1 \end{pmatrix}, \begin{pmatrix} i \\ 1 \end{pmatrix}\right\}$

Out[161]= $\left\{\begin{pmatrix} -i \\ 1 \end{pmatrix}, \begin{pmatrix} i \\ 1 \end{pmatrix}\right\}$

Out[162]= {Cos[θ] − i Sin[θ], Cos[θ] + i Sin[θ]}

Misc